

طریقه استفاده از کنترل Ms-Agent در VB



مایکروسافت کنترلی را ارائه داده است که بسیار شبیه به دستیار آفیس می باشد و به شدت طراحی ظاهر برنامه ها را بهبود می بخشد . مشخصات آن به صورت زیر است :

۱- تبدیل متن به صوت

۲- تشخیص صوت


در ابتدا بهتر است اطمینان حاصل کنیم که آیا این کنترل و ملحقات آن روی کامپیوتر نصب شده است یا خیر . اگر از ویندوز میلیونوم و یا ۲۰۰۰ (و یا بالاتر!) استفاده می کنید این کار به صورت خودکار صورت گرفته است و در مسیر `C:\Windows\msagent\` نصب شده - است . در غیر اینصورت باید آنرا از سایت مایکروسافت تهیه کنید ! (احتمالاً روی CD نصب مرورگر وب مایکروسافت می توانید آنرا پیدا نمایید.)

تا کنون مایکروسافت سه شخصیت (کاراکتر) همراه این کنترل ارائه داده است : Genie , Merlin و Robby. اگر کنترل روی کامپیوتر شما نصب شده باشد مرلین را می توانید در مسیر : `C:\WINDOWS\MSAGENT\CHARS` پیدا کنید. روی فایل `Merlin.acs` در دایرکتوری فوق کلیک راست کنید و گزینه خواص را از منوی ظاهر شده انتخاب نمایید. در صفحه نمایان شده روی برگه `Version` کلیک کنید . در اینجا می توانید مرلین و توضیحات مختصری در مورد او را ببینید .



البته اگر مایل به دریافت کاراکترهای بیشتری هستید می توانید به سایت های ذکر شده در قسمت منابع و مآخذ سری بزنید. در ضمن یک ویرایشگر کاراکتر نیز برای تولید کاراکتر- های مورد علاقه خودتان نیز وجود دارد .

برای استفاده از این Active-X (که قابلیت اجرا در صفحات وب را نیز دارا است) کافی است از منوی Projects گزینه Components... را انتخاب کرده و از لیست ارائه شده Ms Agent را انتخاب کنید و سپس آنرا به فرم برنامه خود اضافه نمایید.



سپس روی فرم دو بار کلیک کنید و در روال رخداد آن بنویسید :

```
Private Sub Form_Load()  
    Dim Assistant As IAgentCtlCharacterEx
```

```

Agent1.Characters.Load "Assistant", _
    "C:\windows\msagent\chars\merlin.acs"
Set Assistant = Agent1.Characters("Assistant")
Assistant.Show
Assistant.Speak "Salaam! Look! I can speak!"
Assistant.MoveTo 320, 240, 10
Assistant.Think "I'm thinking....hmm...E=MC2..!"
End Sub

```



سپس برنامه را اجرا کنید. در کد فوق فرض شده است که مسیر ویندوز C:\windows می باشد. اما این مسیر روی کامپیوترهای مختلف متفاوت است. برای بدست آوردن مسیر ویندوز می توان از کد زیر استفاده کرد:

```

Declare Function GetWindowsDirectory Lib "kernel32" _
    Alias "GetWindowsDirectoryA" ( _
    ByVal lpBuffer As String, ByVal nSize As Long) As Long

Private Declare Function lstrlen Lib "kernel32" Alias _
    "lstrlenA" (ByVal lpString As String) As Long

Function WinDir() As String
    Dim buff As String
    buff = Space(256)
    GetWindowsDirectory buff, Len(buff)
    WinDir = StrZToStr(buff) 'removing null char.
End Function

Function StrZToStr(s As String) As String
    StrZToStr = Left$(s, lstrlen(s))
End Function

```

پویا نمایی کاراکتر (همان انیمیشن خودمان!):



همراه هر کاراکتر یک سری انیمیشن هم ارائه شده است . برای مثال روی فرم برنامه دوبار روی کنترل گماشته مایکروسافت کلیک کنید و در روال رخداد مربوط به کلیک آن بنویسید :

```
Private Sub Agent1_Click(ByVal CharacterID As String, _  
    ByVal Button As Integer, _  
    ByVal Shift As Integer, ByVal x As Integer, _  
    ByVal y As Integer)  
    Agent1.Characters.Character("Assistant").Play "surprised"  
    Agent1.Characters.Character("Assistant").Play "Congratulate_2"  
End Sub
```

لیست انیمیشن‌های موجود برای کاراکتر (البته ممکن است بعضی از کاراکترها بعضی از آنها را پشتیبانی نکنند).

Acknowledge	GlanceUp
Alert	Greet
AlertReturn	GreetReturn
Announce	Hear_1
AnnounceReturn	Hear_2
Blink	Hear_3
Confused	Hear_4
ConfusedReturn	Hide
Congratulate	Idle1_1
CongratulateReturn	Idle1_2
Congratulate_2	Idle1_3
Decline	Idle1_4
DeclineReturn	Idle2_1
DoMagic1	Idle2_2
DoMagic2	Idle3_1
DoMagicReturn	Idle3_1Return
DontRecognize	Idle3_2
DontRecognizeReturn	LookDown
Explain	LookDownBlink
ExplainReturn	LookDownReturn
GestureDown	LookLeft
GestureDownReturn	LookLeftBlink
GestureLeft	LookLeftReturn
GestureLeftReturn	LookRight
GestureRight	LookRightBlink
GestureRightReturn	LookRightReturn
GestureUp	LookUp
GestureUpReturn	LookUpBlink
GetAttention	LookUpReturn
GetAttentionReturn	MoveDown
GlanceDown	MoveDownReturn
GlanceLeft	MoveLeft
GlanceRight	MoveLeftReturn
StopListening	MoveRight
StopListeningReturn	MoveRightReturn
Suggest	MoveUp
SuggestReturn	MoveUpReturn
Surprised	Pleased
SurprisedReturn	PleasedReturn
Think	Processing*
ThinkReturn	ProcessingReturn
Uncertain	Read

UncertainReturn	ReadContinued
Wave	ReadReturn
WaveReturn	Reading*
Write	ReadingReturn
WriteContinued	RestPose
WriteReturn	SadReturn
Writing*	Searching*
WritingReturn	SearchingReturn
Show	StartListening
Sad	StartListeningReturn

در لیست فوق آنهایی که با (*) مشخص شده‌اند انیمیشن‌هایی هستند که تا دستور زیر را اجرا نکنید متوقف نمی‌شوند:

Assistant.Stop

و البته تعداد زیادی از آنها هم دارای کلمه Return هستند که کار فوق را انجام می‌دهد.

برچسب‌های نطق گماشته میکروسافت:



با استفاده از برچسب‌های زیر می‌توان طریقی را که حروف ادا می‌شوند را تغییر داد:

(۱) تاکید روی کلمه: \emp\

(۲) نجوا کردن: "\Chr=""Whisper""\

(۳) تلفظ Dot در آدرس پست الکترونیکی: "\Ctx=""Address\

مثال:

یک دکمه روی فرم جاری برنامه‌ای که تاکنون نوشته ایم قرار دهید. روی آن دوبار کلیک کنید و در روال رخداد مربوط به کلیک آن بنویسید:

```
Private Sub Command1_Click ()
Agent1.Characters.Character("Assistant" _
).Speak "I will emphasize this word."
Agent1.Characters.Character("Assistant" _
).Speak "I will emphasize \emp\this word."
Agent1.Characters.Character("Assistant" _
).Speak "\Chr=""Whisper""\This is an example of whispering."
Agent1.Characters.Character("Assistant" _
).Speak "\Ctx=""Address\webmaster@abstractvb.com."
End Sub
```

هر کدام از برچسب‌ها باید داخل " " قرار گیرند بدین منظور کافی است دو کوتیشن در

اطراف آن بکار برید تا VB متوجه این موضوع بشود.



نکته :

سعی کنید کد زیر را اجرا کنید :

```
Agent1.Characters.Character("Assistant"  
).Speak "HELLO WORLD!"
```

احتمالاً گماشته تک تک حروف را خواهد خواند. برای رفع این مشکل همواره کلمات را با حروف کوچک ارائه دهید.

چوب الف های (Bookmarks) نطق گماشته مایکروسافت :

یکی دیگر از برچسب ها "Mrk" می باشد. هنگامیکه گماشته در حال خواندن متنی می باشد اگر به این برچسب برسد رخداد MSAgent_Bookmark فراخوانی خواهد شد.
مثال :

برنامه جدیدی را شروع کنید و به ترتیبی که گفته شد کنترل گماشته مایکروسافت را روی فرم برنامه قرار دهید. کد کامل مربوط به این فرم را در زیر مشاهده می نمایید :

Option Explicit

Private Merlin As IAgentCtlCharacterEx

Private Sub **Form_Load**()

'Load the Merlin Character

Agent1.Characters.Load "Merlin", _

"c:\windows\msagent\chars\Merlin.acs"

Set Merlin = Agent1.Characters("Merlin")

'Move the Character to the left

Merlin.Left = 500

'Show Merlin without playing the animation

Merlin.Show True

'Use bookmarks in the text to identify where merlin is

Merlin.Speak "\Mrk=1\I am now reading line number one. " & _

"\Mrk=2\Now I am reading line two. " & _

"\Mrk=3\And finally, I am reading line three. \Mrk=4\ "

End Sub

Private Sub **Agent1_Bookmark**(ByVal BookmarkID As Long)

'When Merlin hits the point in the text above where

'there is a bookmark tag this event is fired and the

'ID in the Mrk tag is passed into here as the

'Bookmark ID

Select Case BookmarkID

Case 1

```

        MsgBox "Merlin is currently Reading Line 1", vbInformation
    Case 2
        MsgBox "Merlin is currently Reading Line 2", vbExclamation
    Case 3
        MsgBox "Merlin is currently Reading Line 3", vbCritical
    Case 4
        MsgBox "Merlin has finished Reading all lines", vbQuestion
End Select

End Sub

```

اضافه کردن فرمان (Commands) به کنترل گماشته مایکروسافت :



اگر در هنگام اجرای برنامه روی مرلین کلیک راست کنید منویی ظاهر خواهد شد که فقط در آن گزینه Hide وجود دارد. به این منو گزینه های دیگری نیز می توان افزود. در کد زیر، [...] به معنی است که کلماتی که قبل و یا بعد از فرمان توسط کاربر تلفظ شوند در نظر گرفته نمی شود (تشخیص گفتار). برای مثال در کد زیر اگر Open notepad please توسط کاربر تلفظ شود خروجی در رخداد Command گماشته فقط notepad خواهد بود.

```

AgentChar.Commands.Add "notepad",
    "notepad", "[...] notepad [...]", True, True

```

البته آنچه را که گماشته برمی گرداند فقط حدس او در مورد میزان اطمینان به تشخیص گفتار شما می باشد (confidence). برای مثال از این موضوع با استفاده از کد زیر می - توان اطمینان حاصل نمود:

```

Private Sub MsAgent_Command (ByVal UserInput As Object)
    If UserInput.Confidence > -50 Then
        MsgBox UserInput.Name
    Else
        MsgBox "I have " & UserInput.Confidence & _
            " confidence in that command"
    End If
End Sub

```

مثالی کامل !:

برنامه جدیدی را شروع کنید و کنترل گماشته مایکروسافت را روی فرم برنامه قرار دهید .
کد کامل مربوط به این فرم را در زیر مشاهده می نمایید :

Option Explicit

Private Merlin As IAgentCtlCharacterEx

Private Sub **Form_Load()**

```
Dim sUserName As String
Agent1.Characters.Load "Merlin", _
    "c:\windows\msagent\chars\Merlin.acs"
Set Merlin = Agent1.Characters("Merlin")

Merlin.Balloon.FontName = "verdana"
Merlin.MoveTo 500, 280

Merlin.Commands.Caption = "Merlin" & "'s commands"
Merlin.Commands.Visible = True
Merlin.Commands.Voice = "Merlin" & " commands"

Merlin.Commands.Add _
    "acDate", "Date", "(what date [is it] | ...date...)", True, True
Merlin.Commands.Add _
    "acTime", "Time", "(what time [is it] | ...time...)", True, True
Merlin.Commands.Add _
    "acProcessing", "Processing", "processing", True, True
Merlin.Commands.Add _
    "acReading", "Reading", "reading", True, True
Merlin.Commands.Add _
    "acSearching", "Searching", "searching", True, True
Merlin.Commands.Add _
    "acWriting", "Writing", "writing", True, True

Merlin.Show
Merlin.Play "Announce"
Merlin.Play "Greet"
sUserName = InputBox("Please write your name here :", _
    "Assistant")

Merlin.Speak "Hi! Dear " + sUserName + _
    " . Welcome to VB Tips And Tricks." + _
    "by: Vahid_Nasiri@Yahoo.com "
```

End Sub

Sub **Agent1_Command**(ByVal userInput As Object)

On Error Resume Next

Dim reqHideBalloon

Merlin.Stop

Select Case userInput.Name

Case "acDate":

Set reqHideBalloon = Merlin.Speak(_

```
        "The date is " & Date & ".")
    Case "acTime":
        Set reqHideBalloon = Merlin.Speak( _
            "The current time is " & Time)
    Case "acProcessing":
        Merlin.Play "Processing"
    Case "acReading":
        Merlin.Play "Reading"
    Case "acSearching":
        Merlin.Play "Searching"
    Case "acWriting":
        Merlin.Play "Writing"
End Select

End Sub

Private Sub Form_Unload(Cancel As Integer)
    On Error Resume Next
    Set Merlin = Nothing
    Agent1.Connected = False
End Sub
```

منابع و مأخذ:

<http://www.Abstractvb.com/>
<http://www.microsoft.com/msagent/>
<http://msdn.microsoft.com/workshop/imedia/agent/agentdl.asp>
<http://www.download.com/>
<http://www.msagentring.org/>